An Improved FORTRAN Program for Calculating Modulation Transfer Functions: Concise Communication

Anthony R. Benedetto and Martin L. Nusynowitz

William Beaumont Army Medical Center, El Paso, Texas

An improved* FORTRAN II *program for calculating modulation transfer functions (MTFs) is presented. The program features* (A) *simplified input-data specifications;* (B) *a conversational mode of use; and* (C) *graphic printout of the MTF curve.

J Nucl Med 18: 85–86, 1977

The modulation transfer function (MTF) is widely accepted as the best index of resolution in an imaging system (1–5). Briefly stated, an "infinitely thin, infinitely long" line source of radioactive material is imaged, and the line-source response [i.e., the line spread function (LSF)] of the imaging system is stored for computer analysis. A computer program computes the Fourier transform of the LSF data and outputs the MTF as a function of spatial frequency (*6*). Modulation transfer functions may then be used to compare conveniently and accurately the resolutions of imaging systems. This paper presents a description of an improved computer program for calculating the MTF from LSF data.

DISCUSSION

The generalized LSF curve has its maximum value at the origin of an arbitrary orthogonal coordinate system and is symmetric about the Y axis. If the LSF values are represented by $f(X_I)$ and the displacements along the abscissa as X_I, the modulation transfer function is given by

$$MTF(\nu) = \frac{\sum_{I=-m}^{m} f(X_I) \cos 2\pi\nu X_I}{\sum_{I=-m}^{m} f(X_I)}. \quad (1)$$

The symmetry of the LSF curve implies that an odd number of data points are to be used: the central peak value and an equal number of values on either side of the origin. Note also that the LSF peak value is taken to be exactly at the origin.

Cradduck (*2*) has previously reported a FORTRAN program to calculate MTFs using a large batch-process computer. Cradduck's program has been widely used, but always with (unspecified) modifications (*7–8*). We have improved Cradduck's program in four respects: (A) the specifications of input data are simplified and clarified; (B) the program is conversational, for use on interactive minicomputers; (C) advantage is taken of all the input data, obviating the need for discarding one datum as in Cradduck's program; and (D) a provision is made for plotting the MTF curve as a histogram.

PROGRAM DESCRIPTION

This OS/8* FORTRAN II program runs on a PDP-8/L computer with 32K memory and RK8E disk. A listing of the program is shown in Appendix 1 and a sample run is shown in Appendix 2. The program requests the necessary input, calculates the MTF for the specified frequencies, and prints out a histogram of the MTF as a function of spatial frequency. In our program an odd number of LSF values are entered into an array called $Y(=f(X_I))$

Received July 19, 1976; original accepted Aug. 12, 1976.

For reprints contact: Martin L. Nusynowitz, Nuclear Medicine–Endocrine Service, William Beaumont Army Medical Center, El Paso, TX 79920.

and all elements of Y are used in the subsequent calculations.

This program is easy to use, executes quickly, and generates both tabular and graphic MTF data. The histograms have proven useful in making quick comparisons of various collimator–source distance combinations and in accomplishing quality control checks of scintillation camera performance.

FOOTNOTE

* Digital Equipment Corp., Maynard, Mass.

REFERENCES

1. GOTTSCHALK A, BECK RN, eds.: *Fundamental Problems in Scanning*. Springfield, Ill., C. C. Thomas, 1968, p 22
2. CRADDUCK TD: Assessing performance of radioisotope scanners: Data acquisition. *J Nucl Med* 9: 210–217, 1968
3. GREGG EC: Modulation transfer function, information capacity and performance criteria of scintiscans. *J Nucl Med* 9: 116–127, 1968
4. FREEDMAN GS, ed.: *Tomographic Imaging in Nuclear Medicine*. New York, Society of Nuclear Medicine, 1973, p 134
5. HOFFER PB, BECK RN, GOTTSCHALK A, eds.: *Semiconductor Detectors in the Future of Nuclear Medicine*. New York, Society of Nuclear Medicine, 1971, pp 98–104
6. NUSYNOWITZ ML: *A Clinician's Guide to the Modulation Transfer Function*. Houston, Baylor College Press, 1974
7. GRAHAM LS, POE ND, MACDONALD NS: Collimation for imaging the myocardium. *J Nucl Med* 17: 401–404, 1976
8. GOLDSMITH WA, NUSYNOWITZ ML: Determination of the modulation transfer function (MTF) using a programmable calculator. *Am J Roentgenol Radium Ther Nucl Med* 112: 806–811, 1971

APPENDIX 1. FORTRAN COMPUTER PROGRAM FOR CALCULATING MODULATION TRANSFER FUNCTION FROM LINE-SOURCE DATA. THE PROGRAM ALSO PLOTS A HISTOGRAM OF MTF VERSUS SPATIAL FREQUENCY

```
      DIMENSION X(64), Y(64)
54    READ(1,200)N
200   FORMAT('ENTER THE (ODD) NUMBER OF LSF VALUES TO BE READ,
     & I2 FORMAT:  ', I2)
      READ(1,201)CM
201   FORMAT('ENTER THE CM/CHANNEL (LESS THAN 1.0),F5.3 FORMAT:
     &  ',F5.3)
53    READ(1,202)FREQ
202   FORMAT('ENTER FREQUENCY INCREMENT,F5.3 FORMAT:  ',F5.3)
52    READ(1,203)XMAX
203   FORMAT('ENTER MAXIMUM FREQUENCY TO BE CALCULATED,F5.3 FOR
     &MAT:  ',F5.3)
51    WRITE(1,300)
300   FORMAT('ENTER THE LSF VALUES,F7.1 FORMAT')
1     CONTINUE
      DO 10 I=1,N
10    READ(1,204) Y(I)
204   FORMAT(F7.1)
      XJ = -N/2 - 1
      SUMY = 0.0
      DO 11 I=1,N
      XI = I
      X(I) = (XJ+XI)*CM
11    SUMY = SUMY + Y(I)
      XFREQ = 0.0
      TWOPI = 6.2832
      WRITE(1,301)
301   FORMAT(///,1X,'FREQ',3X,'MTF',2X,51('.'))
2     CONTINUE
      XNUM = 0.0
      DO 12 I=1,N
12    XNUM = XNUM + Y(I)*COS(TWOPI*XFREQ*X(I))
      XMTF = XNUM/SUMY
      IF(XMTF)98,22,22
22    WRITE(1,302)XFREQ,XMTF,
302   FORMAT(1X,F5.3,1X,F5.3)
      SPACE = XMTF*50.0
      ISPACE = SPACE + 0.5
      IF(ISPACE-1)66,77,88
66    WRITE(1,303)
303   FORMAT(' *',/)
      GO TO 3
77    WRITE(1,304)
304   FORMAT(' .*',/)
      GO TO 3
88    LIMIT = ISPACE-1
      I = 0
      WRITE(1,305)(I, K=1,LIMIT),
      WRITE(1,306)
305   FORMAT(' .',50I1)
306   FORMAT('*',/)
3     CONTINUE
      XFREQ = XFREQ + FREQ
      IF(XFREQ - XMAX)2,2,98
98    CONTINUE
      WRITE(1,205)
205   FORMAT(///,'WOULD YOU LIKE TO MAKE FURTHER CALCULATIONS?'
     &,/,'ENTER 1 TO CHANGE ONLY THE LSF VALUES',/,'ENTER 2 TO C
     &HANGE ONLY THE MAXIMUM FREQUENCY',/,'ENTER 3 TO CHANGE THE
     & FREQUENCY INCREMENT',/,'ENTER 4 TO RESTART WITH ALL NEW V
     &ALUES',/)
      READ(1,307) MORE
307   FORMAT('ENTER 5 IF FINISHED:  ',I1)
      GO TO (51,52,53,54,99)MORE
99    CALL EXIT
      END
```

APPENDIX 2. SAMPLE OUTPUT OF COMPUTER PROGRAM, SHOWING DATA ENTRY, LISTING OF FREQUENCY AND MTF VALUES, AND THE HISTOGRAM PLOT OF MTF (HORIZONTAL AXIS) VERSUS SPATIAL FREQUENCY (VERTICAL AXIS)

```
ENTER THE (ODD) NUMBER OF LSF VALUES TO BE READ, I2 FORMAT:  19
ENTER THE CM/CHANNEL (LESS THAN 1.0),F5.3 FORMAT:  .357
ENTER FREQUENCY INCREMENT,F5.3 FORMAT:  .025
ENTER MAXIMUM FREQUENCY TO BE CALCULATED,F5.3 FORMAT:  1.0
ENTER THE LSF VALUES,F7.1 FORMAT
100.
184.
397.
705.
1168.
1882.
2749.
3376.
3726.
3762.
3450.
2880.
2155.
1524.
949.
600.
319.
138.
78.



 FREQ   MTF  ...................................................
 0.000 1.000 .0000000000000000000000000000000000000000000000000*

 0.025 0.984 .000000000000000000000000000000000000000000000000*

 0.050 0.939 .0000000000000000000000000000000000000000000000*

 0.075 0.868 .000000000000000000000000000000000000000000*

 0.100 0.777 .00000000000000000000000000000000000000*

 0.125 0.673 .000000000000000000000000000000000*

 0.150 0.565 .000000000000000000000000000*

 0.175 0.458 .0000000000000000000000*

 0.200 0.360 .00000000000000000*

 0.225 0.274 .0000000000000*

 0.250 0.202 .000000000*

 0.275 0.145 .000000*

 0.300 0.102 .0000*

 0.325 0.070 .000*

 0.350 0.047 .0*

 0.375 0.031 .0*

 0.400 0.019 .*

 0.425 0.010 .*

 0.450 0.004 *
```